

Part C: 螢幕和鍵盤操作

第八章 基本動作

第九章 螢幕控制

第十章 鍵盤控制

第八章 基本動作

設游標位置

清螢幕

顯示文字

鍵盤輸入

本章介紹的系統呼叫

INT 10H

Function 02H: 設游標位置

Function 06H: 滾動螢幕

INT 21H

Function 02H: 顯示一個字

Function 09H: 顯示一段字

Function 0AH: 鍵盤輸入一串字

Function 3FH: 檔案(裝置)輸入

Function 40H: 檔案(裝置)輸出

設游標位置

INT 10H Function 02H

呼叫前, BH放頁(Page)號, DH放列(Row)號,
DL放行(Column)號

```
MOV    AH, 02H
MOV    BH, 0
MOV    DH, 8
MOV    DL, 15
INT    10H
```

清螢幕


使用 INT 10H Function 06H 滾動螢幕功能
滾掉整個螢幕

INT 10H Function 06H

呼叫前, AL放滾動列數; BH放屬性(字和背景顏色); CH放左上角列(Row)號, CL放左上角(Column)號; DH放右下角列(Row)號, DL放右下角(Column)號

字的屬性 (Attribute)

八個位元, 指示字和背景的顏色



課本156頁

顏色	I	R	G	B	Hex	顏色	I	R	G	B	Hex
黑	0	0	0	0	0	灰	1	0	0	0	8
藍	0	0	0	1	1	亮藍	1	0	0	1	9
綠	0	0	1	0	2	亮綠	1	0	1	0	A
靛	0	0	1	1	3	亮靛	1	0	1	1	B
紅	0	1	0	0	4	亮紅	1	1	0	0	C
紫	0	1	0	1	5	亮紫	1	1	0	1	D
棕	0	1	1	0	6	黃	1	1	1	0	E
白	0	1	1	1	7	亮白	1	1	1	1	F

清整個螢幕

```
MOV     AX,0600H
MOV     BH,71H  ;白底藍字
MOV     CX,0
MOV     DX,184FH
INT     10H
```

顯示一段字

Int 21H Function 09H

呼叫前, DX指到字串的開始位址; 字串以錢號
('\$')作為結束, 錢號不顯示出來

```
CUST_MSG  DB  'Customer name?$'
...
MOV     AH,9
LEA     DX,CUST_MSG
INT     21H
```

例子: 顯示ASCII字元集 主程式

```
title a08disas (COM) display ASCII character set
.model small
.code
.org 100h
begin: jmp short a10main
aschar db 00,'$' ;display character
; main procedure:
; -----
.286
a10main proc near
        call b10scrn ;clear screen
        call c10curs ;set cursor
        call d10disp ;display characters
        mov ax,4c00h ;end
        int 21h ; processing
a10main endp
```

清螢幕

```
; clear screen:
; -----
b10scrn proc near
        pusha ;preserve general registers
        mov ax,0600h ;scroll full screen
        mov bh,7 ;attribute: white on black
        mov cx,0 ;upper left location
        mov dx,184fh ;lower right location
        int 10h ;call bios
        popa ;restore general registers
        ret ;return to caller
b10scrn endp
```

設游標位置在 (8,0)

```
; set cursor to 8,0:
; -----
c10curs proc    near
    pusha                ;preserve general registers
    mov  ah,2            ;request set cursor
    mov  bh,0            ;page number 0
    mov  dx,0800h        ;row 0, column 0
    int  10h             ;call bios
    popa                  ;restore general registers
    ret
c10curs endp
```

顯示字

```
; display ASCII characters:
; -----
d10disp proc    near
    pusha                ;preserve general registers
    mov  cx,256           ;initialize 256 iterations
    lea  dx,aschar        ;initialize address of aschar
d20:    cmp  aschar,08h     ;character below 8?
        jb  d30            ;yes, accept
        cmp aschar,0dh     ;below or equal 0dh?
        jbe d40            ; yes, bypass
d30:    mov  ah,9           ;display ascii aschar
        int  21h
d40:    inc  aschar         ;increment for next character
        loop d20           ;decrement cx, loop nonzero
        popa                ;restore general registers
        ret                ;return to caller
d10disp endp
        end    begin
```

鍵盤輸入一串字

INT 21H Function 0AH

須在資料段準備一個參數表(Parameter List)

```
PARAMETER LIST LABEL BYTE
MAX_LEN      DB    20    ;最多輸入20個字 (包括ENTER)
ACT_LEN      DB    ?     ;實際輸入字數(不包括ENTER)
KB_DATA      DB    20 DUP (' ')
```

呼叫

```
MOV  AH, 0AH
LEA  DX, PARA_LIST
INT  21H
```

PARA_LIST是
個LABEL,
屬性是BYTE

鍵盤輸入例子

由鍵盤輸入 'Wilson<enter>'

Para_list內的資料是

ASCII:	20	6	W	i	l	s	o	n	#
十六進制	14	06	57	69	6c	73	6f	6e	0d

最多20字

實際輸入6字

輸入字超過20個字, 第19字之後不再接受<enter>以外的字
打其它字將回應Beep

程式例子: 輸入和顯示名字 資料定義

```
-----  
.model small  
.stack 64  
-----  
.data  
parlist label byte ;name parameter list:  
maxnlen db 20 ; maximum length of name  
actulen db ? ; no. of characters entered  
kbname db 21 dup (' ') ; entered name  
prompt db 'Name? ', '$'
```

主程式

```
.code  
.386  
a10main proc far  
    mov ax,@data ;initialize segment  
    mov ds,ax ; registers  
    mov es,ax  
    call q10clr ;clear screen  
a20:  
    mov dx,0 ;set cursor to 0,0  
    call q20curs  
    call b10inpt ;provide for input of name  
    call q10clr ;clear screen  
    cmp actulen,0 ;name entered?  
    je a30 ; no, exit  
    call c10cntr ;set bell and '$'  
    call d10disp ;center, display name  
    jmp a20  
a30:  
    mov ax,4c00h ;end processing  
    int 21h  
a10main endp
```


等待輸入

```
; display prompt and accept input of name:
; -----
b10inpt proc    near
    push ax            ;preserve used registers
    push dx
    mov  ah,9          ;request display
    lea  dx,prompt
    int  21h
    mov  ah,0ah        ;request keyboard
    lea  dx,parlist; input
    int  21h
    pop  dx            ;restore registers
    pop  ax
    ret
b10inpt endp
```

計算和移動游標位置

```
; set bell and '$' delimiter and
; set cursor at center of screen:
; -----
c10cntr proc    near
    movzx  bx,actulen;replace 0dh with 07h
    mov  kbname[bx],07
    mov  kbname[bx+1],'$' ;set display delimiter
    mov  dl,actulen;locate center column:
    shr  dl,1          ; divide length by 2
    neg  dl            ; reverse sign,
    add  dl,40         ; add 40
    mov  dh,12         ;center row
    call q20curs      ;set cursor
    ret
c10cntr endp
```

顯示名字

```
; display centered name:
; -----
d10disp proc near
    mov     ah,9
    lea    dx, kname    ;display name
    int    21h
    ret
d10disp endp
```

清螢幕

```
; clear screen and set attribute:
; -----
q10clr   proc near
    pusha                ;preserve general registers
    mov    ax,0600h      ;request scroll screen
    mov    bh,30         ;color attribute
    mov    cx,0          ;upper left location
    mov    dx,184fh      ;lower right location
    int    10h          ;call bios
    popa                 ;restore general registers ??
    ret                 ;return to caller
q10clr   endp
```

設游標位置

```
; set cursor to row/column:
; -----
q20curs proc    near ;dx set on entry
    mov  ah,2      ;request set cursor
    mov  bh,0      ;page number 0
    int  10h      ;call bios
    ret
q20curs endp
    end  a10main
```

顯示一個字

INT 21H Function 02H

暫存器 DL 放欲顯示字的ASCII碼

```
MOV     AH,2
MOV     DL,char
INT     21H
```

例子: 顯示一段文字

```
CO_TITLE  DB   'Intertech Corp.',13,10
...
MOV  AH,2
MOV  CX,17
LEA  DI,CO_TITLE
L10:
MOV  DL,[DI]
INT  21H
INC  DI
LOOP L10
...
```

使用 INT 21H Function 40H 作螢幕輸出

暫存器CX放字數, BX放file handle, DX指到文字的開頭地址

螢幕的file handle是01, 列表機是04

輸出不成功將設CF=1, 錯誤碼放在AX

```
PROMPT   DB   'Part number?', 0DH,0AH
...
MOV  AH,40H
MOV  BX,01
MOV  CX,14
LEA  DX,PROMPT
INT  21H
```

使用 INT 21H Function 3FH 作鍵盤輸入

暫存器CX放字數, BX放file handle, DX指到輸入
緩衝區的開頭地址

鍵盤的file handle是00

輸入成功將清除CF, 字數放在AX

```
KNINPUT          DB    20 DUP ( ' ' )
...
MOV    AH, 3FH
MOV    BX, 0
MOV    CX, 20
LEA   DX, KBINPUT
INT    21H
```